



**Clinical Quality Language (CQL) Training
for Measure Developers
Wednesday April 27, 2016
3:00 PM EDT**


Deborah Krauss
Centers for Medicare & Medicaid Services

Bryn Rhodes
ESAC

Agenda

- Welcome and Background
- Quality Data Model (QDM) and CQL
- Retrieval and Filtering
- Relationships and Comparisons
- Building a Measure

Evolving eCQM Standards



Definitions:

HQMF – Health Quality Measure Format

CQL – Clinical Quality Language

QDM – Quality Data Model

Differences Between the Quality Data Model (QDM) Now and When using Clinical Quality Language (CQL)

QDM Now

- Data Model and Logic are both in the QDM

QDM with CQL

- The Data Model will continue to exist as the QDM
- CQL will provide the logic expressions and will replace that function currently in the QDM

Benefits of CQL

	QDM Logic	CQL Logic
Modularity and Computability	Low	High
Data Model Flexibility	**	High
Expressive and Robust Logic Expression	Low	High
Duplicative work for Implementers, Vendors, and Developers	Yes	Lower

Proposed Timeline For Updating Standards

Work Effort: 2016 through Fall 2017

Fall 2017 +

Measures using QDM v4.2 & HQMF 2.1

Measure Development

- 2015
- 2016

Testing CQL – QDM – HQMF 2.1

Testing and Development

- Measure Developers
- Implementers & Vendors
- CQL Training/Education
- Measure Authoring Tool
- Bonnie & Cypress
- Quality Data Model
- Integration Testing
- Feedback Loops

**Testing eCQM using CQL -
– QDM – HQMF 2.1**

Measure Development and Testing in a simulated environment

- Starts 2017

BUILDING MEASURES WITH CQL AND QDM


Clinical Quality Language (CQL)

- Health Level 7(HL7) standard designed to
 - Enable automated point-to-point sharing of executable clinical knowledge
 - Provide a clinically focused, author-friendly, and human-readable language
- Developed as part of a larger Clinical Quality Measure – Clinical Decision Support (CQM-CDS) harmonization effort (Clinical Quality Framework [CQF])
- Currently a Draft Standard for Trial Use (DSTU) publication

Quality Data Model (QDM)

- Describes clinical concepts in a standard format
 - Category – Single clinical concept
 - e.g., Medication, Procedure, Encounter
 - Datatype – Context of use for each category
 - e.g., “Medication, Active”, “Medication, Administered”
 - Attribute – Provides specific detail about an element
 - e.g., admissionDatetime, result
 - QDM Element – Specific category and associated datatype
 - e.g., “Encounter, Performed: Inpatient Encounter”

QDM Elements in CQL



CQL Retrieval

```
58  
59  
60 ["Encounter, Performed": "Inpatient"]
```

- Returns all encounters whose *code* attribute is a member of the “Inpatient” value set
- *code* is the default for retrieves

```
60  
61  
62 ["Encounter, Performed": diagnosis in "Pharyngitis"]
```

- Returns all encounters whose *diagnosis* attribute is a member of the “Pharyngitis” value set

CQL Retrieval Example

58

59

60

```
["Encounter, Performed": "Inpatient"]
```

id	typeCode	diagnosis	admissionDateTime	dischargeDateTime
1	INP-01	PHRYNG	1/1/2015 3:33:00 PM	1/5/2015 10:11:00 AM
2	INP-02	PHRYNG	2/1/2015 3:33:00 PM	2/5/2015 10:11:00 AM
3	INP-03	ASTHMA	3/1/2015 3:33:00 PM	3/5/2015 10:11:00 AM

60

61

62

```
["Encounter, Performed": diagnosis in "Pharyngitis"]
```

id	typeCode	diagnosis	admissionDateTime	dischargeDateTime
1	INP-01	PHRYNG	1/1/2015 3:33:00 PM	1/5/2015 10:11:00 AM
2	INP-02	PHRYNG	2/1/2015 3:33:00 PM	2/5/2015 10:11:00 AM

Negation

- Negation is indicated in QDM with the *negationRationale* attribute

```
2
3 /* Encounters that were performed */
4 ["Encounter, Performed": "Inpatient"] Encounter
5   where Encounter.negationRationale is null
6
7 /* Encounters that were not performed */
8 ["Encounter, Performed": "Inpatient"] Encounter
9   where Encounter.negationRationale is not null
10
11 /* Encounters that were not performed for a specific reason */
12 ["Encounter, Performed": "Inpatient"] Encounter
13   where Encounter.negationRationale in "Encounter Not Done Reason"
14
```

- Default retrieves encounters with and without a *negationRationale*, so the intended result must be specified with a filter

Negation Examples

```

2
3 /* Encounters that were performed */
4 ["Encounter, Performed": "Inpatient"] Encounter
5   where Encounter.negationRationale is null
6

```

id	typeCode	diagnosis	admissionDateTime	dischargeDateTime	negationRationale
1	INP-01	PHRYNG	1/1/2015 3:33:00 PM	1/5/2015 10:11:00 AM	<No Value>
2	INP-02	PHRYNG	2/1/2015 3:33:00 PM	2/5/2015 10:11:00 AM	<No Value>

```

6
7 /* Encounters that were not performed */
8 ["Encounter, Performed": "Inpatient"] Encounter
9   where Encounter.negationRationale is not null
10

```

id	typeCode	diagnosis	admissionDateTime	dischargeDateTime	negationRationale
3	INP-03	ASTHMA	3/1/2015 3:33:00 PM	3/5/2015 10:11:00 AM	PT-REASON

```

10
11 /* Encounters that were not performed for a specific reason */
12 ["Encounter, Performed": "Inpatient"] Encounter
13   where Encounter.negationRationale in "Encounter Not Done Reason"
14

```

Filtering by Existence

QDM filters by existence of *result* attribute

```

2
3 /* "Occurrence A of Laboratory Test, Performed: CD4+ Count (result)" */
4

```

CQL uses a *null* test

```

11
12 define "CD4+ Count With Results and Medications":
13   ["Laboratory Test, Performed": "CD4+ Count"] CD4Count
14   where CD4Count.result is not null
15

```

Note that comparison to *null* is not correct

```

17
18 define "CD4+ Count With Results and Medications":
19   ["Laboratory Test, Performed": "CD4+ Count"] CD4Count
20   where CD4Count.result <> null
21

```

This will not return any results because the answer to a comparison with *null* is “unknown”

Filtering by Existence Example

```

11
12 define "CD4+ Count With Results and Medications":
13   ["Laboratory Test, Performed": "CD4+ Count"] CD4Count
14   where CD4Count.result is not null
15

```

id	typeCode	result
2	CD4+COUNT	10

```

17
18 define "CD4+ Count With Results and Medications":
19   ["Laboratory Test, Performed": "CD4+ Count"] CD4Count
20   where CD4Count.result <> null
21

```

id	typeCode	result
----	----------	--------

CQL Filtering by Value Set

- In addition to filtering within the retrieve, value sets can be used to filter within the *where* clause using the *in* operator

```
26  
27 define "Denominator Exclusions 1":  
28   "Encounter Inpatient" Encounter  
29     where Encounter.facilityLocation in "Neonatal Intensive Care Unit (NICU)"  
30       or Encounter.dischargeStatus in "Patient Expired"  
31       or Encounter.dischargeStatus in "Discharge To Acute Care Facility"  
32
```

CQL Filtering by Value Set

- Single code filters can also be applied using a *code* definition

```
3
4 codesystem "SNOMED-CT": '2.16.840.1.113883.6.96' version '2016A'
5
6 code "Breastfeeding intention": '123456789' system "SNOMED-CT" display 'Breastfeeding intention'
7
8 define "Breastfeeding Intention":
9   ["Risk Assessment, Performed": "Breastfeeding intention"] RiskAssessment
10   where RiskAssessment.negationRationale is null
11
```

Filtering by Values and Quantities

- QDM filter by quantity

```
5  
6 /* "Laboratory Test, Performed: LDL-c Test (result < 100 mg/dL)" */  
7
```

- CQL equivalent

```
49  
50 define "LDL-c Test In Range":  
51   ["Laboratory Test, Performed": "LDL-c Test"] LDLCTest  
52   where LDLCTest.result < 100 "mg/dL"  
53
```

- CQL uses UCUM units to specify quantities

Related Information

- QDM provides temporal relationships

```
2
3  /*
4   "Encounter, Performed: Hospital Inpatient"
5     starts before start of "Diagnosis: Diabetes"
6  */
7
```

- CQL equivalent

```
61
62 define "Valid Encounters":
63   ["Encounter, Performed": "Hospital Inpatient"] Encounter
64   with ["Diagnosis": "Diabetes"] Diagnosis
65   .....
66   such that Encounter.admissionDatetime occurs before Diagnosis.onsetDatetime
```

CQL Date and Time Comparisons

Date and Time comparisons can be precision-based

Operator	Precision-based Operator
=	same <precision> as
<	before <precision> of
>	after <precision> of
<=	same <precision> or before
>=	same <precision> or after

```

69
70 define "Valid Encounters":
71   ["Encounter, Performed": "Hospital Inpatient"] Encounter
72   with ["Diagnosis": "Diabetes"] Diagnosis
73   such that Encounter.admissionDatetime occurs same day as Diagnosis.onsetDatetime
74

```

Hospital inpatient encounters with diabetes diagnoses such that the admission was on the same day as the diagnosis onset

Interval Comparisons

QDM “during”

```
3  
4 "Encounter, Performed: Hospital Inpatient" during "Diagnosis: Diabetes"  
5
```

CQL equivalent

```
81  
82 define "Valid Encounters":  
83   ["Encounter, Performed": "Hospital Inpatient"] Encounter  
84   with ["Diagnosis": "Diabetes"] Diagnosis  
85   such that Interval[Encounter.admissionDatetime, Encounter.dischargeDatetime]  
86     during Interval[Diagnosis.onsetDatetime, Diagnosis.abatementDatetime]  
87
```

CQL Interval Operators

Operator/Inverse	Diagram	Interpretation
X same as Y Y same as X		start of X = start of Y and end of X = end of Y
X before Y Y after X		end of X < start of Y
X meets before Y Y meets after X X meets Y		successor of end of X = start of Y
X overlaps before Y Y overlaps after X X overlaps Y		start of X <= start of Y and start of Y <= end of X
X begins Y		start of X = start of Y and end of X <= end of Y
X included in (during) Y Y includes X		start of X >= start of Y and end of X <= end of Y
X ends Y		start of X >= start of Y and end of X = end of Y

Combining Conditions

- QDM encounters in a range

```
5
6   AND: "Occurrence A of Encounter, Performed: BH Outpatient encounter" >=
7     42 day(s) starts before start of "Measurement End Date"
8   AND: "Occurrence A of Encounter, Performed: BH Outpatient encounter"
9     starts after start of "Measurement Start Date"
10
```

- CQL equivalent

```
97
98 define "Valid Encounters":
99   ["Encounter, Performed": "BH Outpatient encounter"] Encounter
100   where Encounter.admissionDateTime occurs 42 days or less before end of "Measurement Period"
101   and Encounter.admissionDateTime occurs after start of "Measurement Period"
102
```

Combining Conditions (cont)

- Finding follow up visits

```

3
4   AND: "Occurrence A of Encounter, Performed: HIV Visit" during "Measurement Period"
5   AND: "Occurrence B of Encounter, Performed: HIV Visit" during "Measurement Period"
6   AND: "Occurrence B of Encounter, Performed: HIV Visit" >= 90 day(s)
7       starts after end of "Occurrence A of Encounter, Performed: HIV Visit"
8

```

- CQL equivalent

```

109
110 define "HIV Visits During Measurement Period":
111   ["Encounter, Performed": "HIV Visit"] Encounter
112   where Interval[Encounter.admissionDateTime, Encounter.dischargeDateTime] during "Measurement Period"
113
114 define "HIV Followup Visits":
115   "HIV Visits During Measurement Period" FollowupVisit
116   with "HIV Visits During Measurement Period" PriorVisit
117   such that FollowupVisit.admissionDateTime occurs 90 days or more after PriorVisit.dischargeDateTime
118

```

Combining Conditions (cont)

- Medications ordered after a specific lab

```

3
4 AND: "Occurrence A of Medication, Order: Dapsone and pyrimethamine" <=
5   3 month(s) starts after end of "Occurrence A of Laboratory Test, Performed: CD4+ Count (result)"
6 AND: "Occurrent A of Medication, Order: Leucovorin" <= 3 month(s)
7   starts after end of "Occurrence A of Laboratory Test, Performed: CD4+ Count (result)"
8

```

- CQL equivalent

```

126
127 define "CD4+ Count With Results and Medications":
128   ["Laboratory Test, Performed": "CD4+ Count"] CD4Count
129   with ["Medication, Order": "Dapsone and pyrimethamine"] DapsoneOrder
130   such that DapsoneOrder.signedDatetime occurs 3 months or less before CD4Count.stopDateTime
131   with ["Medication, Order": "Leucovorin"] LeucovorinOrder
132   such that LeucovorinOrder.signedDatetime occurs 3 months or less before CD4Count.stopDateTime
133   where CD4Count.result is not null
134

```

Building a Measure – CMS 126v4

- **Initial Population =**
 - AND: Age \geq 5 year(s) at: "Measurement Period"
 - AND: Age $<$ 64 year(s) at: "Measurement Period"
 - AND: "Diagnosis, Active: Persistent Asthma" overlaps "Measurement Period"
 - AND: Union of:
 - "Encounter, Performed: Office Visit"
 - "Encounter, Performed: Face-to-Face Interaction"
 - "Encounter, Performed: Preventive Care - Established Office Visit, 0 to 17"
 - "Encounter, Performed: Preventive Care Services - Established Office Visit, 18 and Up"
 - "Encounter, Performed: Preventive Care Services-Initial Office Visit, 18 and Up"
 - "Encounter, Performed: Preventive Care- Initial Office Visit, 0 to 17"
 - "Encounter, Performed: Home Healthcare Services"
 - during "Measurement Period"
- **Denominator =**
 - AND: Initial Population
- **Denominator Exclusions =**
 - OR: Union of:
 - "Diagnosis, Active: Chronic Obstructive Pulmonary Disease"
 - "Diagnosis, Active: Emphysema"
 - "Diagnosis, Active: Cystic Fibrosis"
 - "Diagnosis, Active: Acute Respiratory Failure"
 - overlaps "Measurement Period"
- **Numerator =**
 - AND: "Medication, Dispensed: Preferred Asthma Therapy" during "Measurement Period"
- **Numerator Exclusions =**
 - None
- **Denominator Exceptions =**
 - None

CMS 126v4 – Initial Population

- **Initial Population =**

- AND: Age \geq 5 year(s) at: "Measurement Period"
- AND: Age $<$ 64 year(s) at: "Measurement Period"
- AND: "Diagnosis, Active: Persistent Asthma" overlaps "Measurement Period"
- AND: Union of:
 - "Encounter, Performed: Office Visit"
 - "Encounter, Performed: Face-to-Face Interaction"
 - "Encounter, Performed: Preventive Care - Established Office Visit, 0 to 17"
 - "Encounter, Performed: Preventive Care Services - Established Office Visit, 18 and Up"
 - "Encounter, Performed: Preventive Care Services-Initial Office Visit, 18 and Up"
 - "Encounter, Performed: Preventive Care- Initial Office Visit, 0 to 17"
 - "Encounter, Performed: Home Healthcare Services"
 - during "Measurement Period"

- 3 main components
 - Demographic
 - Asthma diagnosis
 - Valid encounter

CMS 126v4 – In Demographic

- Patients between 5 and 64 years

```
39  
40 AND: Age >= 5 year(s) at: "Measurement Period"  
41 AND: Age < 64 year(s) at: "Measurement Period"  
42
```

```
2  
3 define "In Demographic":  
4   AgeInYearsAt(start of "Measurement Period") >= 5  
5   and AgeInYearsAt(start of "Measurement Period") < 65  
6
```

CMS 126v4 – Asthma Diagnosis

- Patients with an diagnosis of asthma

```
53  
54 AND: "Diagnosis, Active: Persistent Asthma" overlaps "Measurement Period"  
55
```

```
177  
178 define "Asthma Diagnosis":  
179   ["Diagnosis": "Persistent Asthma"] Diagnosis  
180   where Interval[Diagnosis.onsetDatetime, Diagnosis.abatementDatetime] overlaps "Measurement Period"  
181
```

CMS 126v4 – Valid Encounters

- Encounters during measurement period

```

67
68     AND: Union of:
69         "Encounter, Performed: Office Visit"
70         "Encounter, Performed: Face-to-Face Interaction"
71         "Encounter, Performed: Preventive Care - Established Office Visit, 0 to 17"
72         "Encounter, Performed: Preventive Care Services - Established Office Visit, 18 and Up"
73         "Encounter, Performed: Preventive Care Services-Initial Office Visit, 18 and Up"
74         "Encounter, Performed: Preventive Care- Initial Office Visit, 0 to 17"
75         "Encounter, Performed: Home Healthcare Services"
76     during "Measurement Period"
77

```

```

181
182 define "Valid Encounters":
183     ([ "Encounter, Performed": "Office Visit" ]
184     union [ "Encounter, Performed": "Face-to-Face Interaction" ]
185     union [ "Encounter, Performed": "Preventive Care - Established Office Visit, 0 to 17" ]
186     union [ "Encounter, Performed": "Preventive Care Services - Established Office Visit, 18 and Up" ]
187     union [ "Encounter, Performed": "Preventive Care Services-Initial Office Visit, 18 and Up" ]
188     union [ "Encounter, Performed": "Preventive Care- Initial Office Visit, 0 to 17" ]
189     union [ "Encounter, Performed": "Home Healthcare Services" ]
190 ) E where Interval[E.admissionDateTime, E.dischargeDateTime] during "Measurement Period"
191

```

CMS 126v4 – Initial Population

- Patients in the demographic with a diagnosis of asthma and at least one valid encounter

```
191  
192 define "In Initial Population":  
193     "In Demographic"  
194     and exists "Asthma Diagnosis"  
195     and exists "Valid Encounters"  
196
```

CMS 126v4 - Denominator

- Initial population criteria is implied, so the definition is omitted from the CQL
- Leave a comment instead

```
17  
18 /* Denominator is the same as the initial population */  
19
```

CMS 126v4 – Denominator Exceptions

- Patients with COPD, emphysema, etc.

```
110
111     OR: Union of:
112         "Diagnosis, Active: Chronic Obstructive Pulmonary Disease"
113         "Diagnosis, Active: Emphysema"
114         "Diagnosis, Active: Cystic Fibrosis"
115         "Diagnosis, Active: Acute Respiratory Failure"
116         overlaps "Measurement Period"
117
```

```
2
3 define "Excepted Diagnoses":
4   ([ "Diagnosis": "Chronic Obstructive Pulmonary Disease" ]
5     union [ "Diagnosis": "Emphysema" ]
6     union [ "Diagnosis": "Cystic Fibrosis" ]
7     union [ "Diagnosis": "Actue Repiratory Failure" ]
8   ) Diagnosis
9   where Interval[Diagnosis.onsetDatetime, Diagnosis.abatementDatetime]
10     overlaps "Measurement Period"
11
12 define "Denominator Exceptions":
13   exists "Excepted Diagnoses"
14
```

CMS 126v4 - Numerator

- Patients with preferred asthma therapy

```
132  
133     AND: "Medication, Dispensed: Preferred Asthma Therapy" during "Measurement Period"  
134
```

```
211  
212 define "Preferred Asthma Therapy":  
213   ["Medication, Dispensed": "Preferred Asthma Therapy"] Medication  
214     where Interval[Medication.startDatetime, Medication.stopDateTime] during "Measurement Period"  
215  
216 define "Numerator":  
217   exists "Preferred Asthma Therapy"  
218
```

CMS 126v4 – Criteria

- Criteria definitions for the measure

```
3
4 define "In Initial Population":
5     "In Demographic"
6     and exists "Asthma Diagnosis"
7     and exists "Valid Encounters"
8
9     /* Denominator is the same as the initial population */
10
11 define "Denominator Exceptions":
12     exists "Excepted Diagnoses"
13
14 define "Numerator":
15     exists "Preferred Asthma Therpay"
16
```

Resources

- HL7 Standard: Clinical Quality Language Specification, Release 1 DSTU
 - http://www.hl7.org/implement/standards/product_brief.cfm?product_id=400
- HL7 CDS Workgroup Project Homepage
 - http://wiki.hl7.org/index.php?title=Clinical_Quality_Language
- GitHub Tools Repository
 - https://github.com/cqframework/clinical_quality_language
- eCQI Resource Center CQL Space
 - <https://ecqi.healthit.gov/cql>

Questions?